```
(* ---------------------------------------- *)
(*  rr2.nb   [a473s-6.pdf ]                 *)
(* ---------------------------------------- *)

Clear["Global`*"]

ip[f_, g_] := Integrate[f[x] g[x], {x, 0, a}]

norm[f_] := ip[f, f]


nc[f_] := ip[f, f]^(-1 / 2)

p1[x_] := x (a - x)

p2[x_] := p1[x] (x - a / 2)

p2[x]
```

$$(a - x)\, x \left(-\frac{a}{2} + x\right)$$

```
c1 = nc[p1]
```

$$\frac{\sqrt{30}}{\sqrt{a^5}}$$

```
f1[x_] := c1 p1[x]

c2 = nc[p2]
```

$$\frac{2\sqrt{210}}{\sqrt{a^7}}$$

```
f2[x_] := c2 p2[x]

f1[x]
```

$$\frac{\sqrt{30}\ (a - x)\, x}{\sqrt{a^5}}$$

```
ip[f1', f2']

0

(* Hence for these trial functions 2x2 matrix <n H m>  is diagonal  *)
(* diag elements are variational eigenvalue approximations  *)


ip[f1', f1']
```

$$\frac{10}{a^2}$$

```
10/a^2 (* exact e1 = Pi^2/a^2  *)
```

$$\frac{10}{a^2}$$

```
ip[f2', f2']
```

$$\frac{42}{a^2}$$

```
42/a^2 (* exact e2 = 4 Pi^2 /a^2  *)
```

$$\frac{42}{a^2}$$

```
(* consruct a third function, orthogonal to f2 by symmetry *)
(* about a/2, and adjusted to be orthogonal to f1 by b = -a^2/7  *)

p3[x_] := p1[x] * ((a - 2 x) * (a - 2 x) - a^2 / 7)

ip[p3, p1]
```

0

```
c3 = nc[p3]
```

$$\frac{21\sqrt{\frac{5}{2}}}{\sqrt{a^9}}$$

```
f3[x_] := c3 p3[x]

{ip[f1, f2], ip[f1, f3], ip[f2, f3]}
```

{0, 0, 0}

```
Map[norm, {f1, f2, f3}]
```

{1, 1, 1}

```
{1, 1, 1}
```

{1, 1, 1}

```
(* The functions are o.n and satisfy BC  *)

vf = {f1, f2, f3}
```

{f1, f2, f3}

```
vf[[1]]
```

f1

```
(* choose  a  for numerical work  *)
a = 2
```

2

```
Plot[{f1[x], f2[x], f3[x]}, {x, 0, a}]
```




```
(*  Note: vf[[2]] selects the vector item 2 (sorry about that) *)

h  = Table[ip[vf[[i]]', vf[[j]]'], {i, 3}, {j, 3}]
```

$$\left\{\left\{\frac{5}{2}, 0, \frac{\sqrt{3}}{2}\right\}, \left\{0, \frac{21}{2}, 0\right\}, \left\{\frac{\sqrt{3}}{2}, 0, \frac{51}{2}\right\}\right\}$$

```
MatrixForm[h]
```

$$\begin{pmatrix} \frac{5}{2} & 0 & \frac{\sqrt{3}}{2} \\ 0 & \frac{21}{2} & 0 \\ \frac{\sqrt{3}}{2} & 0 & \frac{51}{2} \end{pmatrix}$$

```
(* The function Eigenvalues cannot work with Matrix Form  *)

Eigenvalues[h]
```

$$\left\{14 + \sqrt{133}, \frac{21}{2}, 14 - \sqrt{133}\right\}$$

```
N[%, 10]
```

```
{25.53256259, 10.50000000, 2.467437405}
```

```
(* Recall the exact eigenvalues for L = a = 2   *)

Table[(i * Pi / a)^2, {i, 3}]
```

$$\left\{\frac{\pi^2}{4}, \pi^2, \frac{9\,\pi^2}{4}\right\}$$

```
N[%, 10]
```

```
{2.467401100, 9.869604401, 22.20660990}
```

```
(* improvement in e1 over the {f1, f2} approximation  *)
(* e2 estimate is unchanged for it is found in the same *)
(*  1-dimensnional odd space (odd about x = a/2).  *)
```